We study the deformation occuring in the system of a sphere rolling on a surface. This is an example treated by J. Hadamard in J. Hadamard, *Sur l'équilibre des plaques élastiques circulaires libres ou appuyées et celui de la sphère isotrope*, Annales scientifiques de l'E. N. S., 3e série, 18 (1901), pp. 313-342.

"Lorsqu'un corps elastique est deforme par l'action de corps rigides qui doivent rester en contact avec lui, le contact ayant lieu sans frottement, la deformation qu'il subit ne peut etre quelconque, quels que soient les corps deformants, puisque la tension a la surface doit etre normale" (pp. 337).

```
> with(Ore_algebra):
> with(OreModules):
```

Introducting the new variable $\rho = \sqrt{x^2 + y^2 + z^2}$, where x, y, z denote the coordinates of a point in the space, J. Hadamard showed on page 336 how the system of partial differential equations corresponding to his problem can be rewritten as a linear system of ordinary differential equations in ρ and $\mathrm{D} = \mathrm{d}/\mathrm{d}\rho$. Therefore, we first define the Weyl algebra $Alg = A_1$, where D is the differential operator w.r.t. ρ. In fact, apart from the constants appearing in the system matrix below, the system involves only the Euler operator $\mathrm{E} = \rho\,\mathrm{D}$. Below we shall also study this system by viewing it over the Ore algebra containing the Euler operator rather than D and ρ.

```
> Alg := DefineOreAlgebra(diff=[D,rho], polynom=[rho], comm=[lambda,mu]):
```

We enter the system matrix R.

```
> R := evalm([[rho*D+1/2, ((lambda+mu)/2)*(rho*D-1), 1/2, 0],
> [2*rho*D, -(3*lambda+2*mu), rho*D+3, 0], [-rho*D, lambda, -1, 2*mu*(rho*D+1)]]);
```

$$R := \begin{bmatrix} \rho\,\mathrm{D} + \frac{1}{2} & \frac{(\lambda+\mu)\,(\rho\,\mathrm{D}-1)}{2} & \frac{1}{2} & 0 \\ 2\,\rho\,\mathrm{D} & -3\,\lambda - 2\,\mu & \rho\,\mathrm{D} + 3 & 0 \\ -\rho\,\mathrm{D} & \lambda & -1 & 2\,\mu\,(\rho\,\mathrm{D}+1) \end{bmatrix}$$

Then, the linear system of ordinary differential equations $R\,z = 0$, where $z = (\theta, \sigma, K, G)^T$, is defined by:

```
> ApplyMatrix(R, [theta(rho),sigma(rho),K(rho),G(rho)], Alg)=evalm([[0],[0],[0]]);
```

$$\begin{bmatrix} \frac{1}{2}\,\theta(\rho) + \rho\,(\frac{d}{d\rho}\,\theta(\rho)) - \frac{1}{2}\,\lambda\,\sigma(\rho) - \frac{1}{2}\,\sigma(\rho)\,\mu + \frac{1}{2}\,\rho\,(\frac{d}{d\rho}\,\sigma(\rho))\,\lambda + \frac{1}{2}\,\rho\,(\frac{d}{d\rho}\,\sigma(\rho))\,\mu + \frac{1}{2}\,\mathrm{K}(\rho) \\ 2\,\rho\,(\frac{d}{d\rho}\,\theta(\rho)) - 3\,\lambda\,\sigma(\rho) - 2\,\sigma(\rho)\,\mu + 3\,\mathrm{K}(\rho) + \rho\,(\frac{d}{d\rho}\,\mathrm{K}(\rho)) \\ -\rho\,(\frac{d}{d\rho}\,\theta(\rho)) + \lambda\,\sigma(\rho) - \mathrm{K}(\rho) + 2\,\mu\,\mathrm{G}(\rho) + 2\,\mu\,\rho\,(\frac{d}{d\rho}\,\mathrm{G}(\rho)) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

Let us compute the formal adjoint of R:

```
> R_adj := Involution(R, Alg);
```

$$R\_adj := \begin{bmatrix} -\rho\,\mathrm{D} - \frac{1}{2} & -2 - 2\,\rho\,\mathrm{D} & \rho\,\mathrm{D} + 1 \\ -\frac{\lambda\,\rho\,\mathrm{D}}{2} - \lambda - \frac{\mu\,\rho\,\mathrm{D}}{2} - \mu & -3\,\lambda - 2\,\mu & \lambda \\ \frac{1}{2} & -\rho\,\mathrm{D} + 2 & -1 \\ 0 & 0 & -2\,\mu\,\rho\,\mathrm{D} \end{bmatrix}$$

In order to check the parametrizability of the system, we compute the first extension module with values in *Alg* of the left *Alg*-module associated with *R_adj*:

```
>  st := time():  Ext1 := Exti(R_adj, Alg, 1):  time()-st; Ext1[1];
```

$$0.601$$

$$\begin{bmatrix} \rho\,\mathrm{D}+1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & \rho\,\mathrm{D}+1 \end{bmatrix}$$

We find that the torsion submodule $\mathrm{t}(M)$ of the left *Alg*-module M which is associated with the system is different from the zero module because *Ext1*[1] is not an identity matrix. A generating set for the torsion submodule $\mathrm{t}(M)$ is given by the rows of *Ext1*[2]:

```
>  Ext1[2];
```

$$\begin{bmatrix} 1 & \lambda+\mu & -1 & -2\,\mu \\ 0 & -\lambda-2\,\mu & \rho\,\mathrm{D}+1 & 4\,\mu\,\rho\,\mathrm{D}+4\,\mu \\ 0 & \lambda\,\rho\,\mathrm{D}+\mu\,\rho\,\mathrm{D}-2\,\mu & 0 & 4\,\mu\,\rho\,\mathrm{D}+6\,\mu \end{bmatrix}$$

The ith entry in the diagonal of *Ext1*[1] corresponds to the ith row of *Ext1*[2], which means that the ith generator of the torsion submodule $\mathrm{t}(M)$ of M is annihilated by the ith element in the diagonal of *Ext1*[1], i.e., we have $(\rho\,\mathrm{D}+1)\,r=0$ in M, where r is the first or the third row in *Ext1*[2] modulo R. Note that the second row r_2 of *Ext1*[2] is zero modulo the system equations, which is clear from the second diagonal entry in *Ext1*[1], namely we find that $1\;r_2$ is zero in M.

Hence, the first and the third row in *Ext1*[2] define autonomous elements of the system and the system $R\,z=0$ is not parametrizable.

However, the torsion-free module $M\;/\;\mathrm{t}(M)$ is parametrizable. A parametrization of $M\;/\;\mathrm{t}(M)$ is given by *Ext1*[3]:

```
>  map(a->map(b->collect(b, D), a), map(factor, Ext1[3]));
```

$$\begin{bmatrix} 2\,\mu\,((\lambda\,\rho^2+\mu\,\rho^2)\,\mathrm{D}^2+(3\,\lambda\,\rho+3\,\mu\,\rho)\,\mathrm{D}-\mu) \\ -2\,\mu\,(\rho\,\mathrm{D}+3)\,(2\,\rho\,\mathrm{D}+1) \\ -2\,\mu\,((2\,\mu\,\rho^2+2\,\lambda\,\rho^2)\,\mathrm{D}^2+(4\,\mu\,\rho+6\,\lambda\,\rho)\,\mathrm{D}+2\,\mu+3\,\lambda) \\ \lambda\,\rho^2\,\mathrm{D}^2+2\,\lambda\,\rho\,\mathrm{D}+\mu\,\rho^2\,\mathrm{D}^2-2\,\mu \end{bmatrix}$$

In other words, we have *Ext1*[2] $u=0 \iff u=$ *Ext1*[3] ξ_1, where u and ξ_1 are respectively a smooth vector and a smooth function, i.e., we have:

```
>  evalm([seq([u[i](rho)], i=1..4)])=Parametrization(Ext1[2], Alg);
```

$$\begin{bmatrix} u_1(\rho) \\ u_2(\rho) \\ u_3(\rho) \\ u_4(\rho) \end{bmatrix} = \begin{bmatrix} 2\,\mu\,(-\mu\,\xi_1(\rho)+\%1\,\lambda\,\rho^2+\mu\,\rho^2\,\%1+3\,\lambda\,\rho\,\%2+3\,\mu\,\rho\,\%2) \\ -2\,\mu\,(3\,\xi_1(\rho)+7\,\%2\,\rho+2\,\%1\,\rho^2) \\ -2\,\mu\,(3\,\xi_1(\rho)\,\lambda+2\,\mu\,\xi_1(\rho)+6\,\lambda\,\rho\,\%2+4\,\mu\,\rho\,\%2+2\,\%1\,\lambda\,\rho^2+2\,\mu\,\rho^2\,\%1) \\ -2\,\mu\,\xi_1(\rho)+2\,\lambda\,\rho\,\%2+\%1\,\lambda\,\rho^2+\mu\,\rho^2\,\%1 \end{bmatrix}$$

$$\%1 := \frac{d^2}{d\rho^2}\,\xi_1(\rho)$$

$$\%2 := \frac{d}{d\rho}\,\xi_1(\rho)$$

In order to find a parametrization of M (i.e., of the linear system of the rolling sphere), we have to "integrate the torsion elements" of the system. We are going to explain what this means (see also A. Quadrat, D. Robertz, *Parametrizing all solutions of uncontrollable multidimensional linear systems*, to appear in the Proceedings of the 16th IFAC World Congress, Prague, 2005). The previous computation of the first extension module with values in *Alg* of the module associated with *R_adj* gives a way to split

R into R" and R' such that the system $R\ z = 0$ is equivalent to R" $\tau = 0$ and $\tau = R'\ z$, where R' is defined by *Ext1* [2] and we can compute R" as follows:

```
> Factorize(R, Ext1[2], Alg);
```

$$\begin{bmatrix} \rho\,\mathrm{D} + \frac{1}{2} & 1 & \frac{-1}{2} \\ 2\,\rho\,\mathrm{D} & 3 & -2 \\ -\rho\,\mathrm{D} & -1 & 1 \end{bmatrix}$$

The idea is now to solve the homogeneous system R" $\tau = 0$ first and afterwards find a particular solution of the inhomogeneous system $R'\ z = \tau$ (which is always possible for linear systems of ODEs).

Instead of R" $\tau = 0$, we consider the equivalent system *Ext1* [1] $\tau = 0$, which is in diagonal form:

```
> Ext1[1];
```

$$\begin{bmatrix} \rho\,\mathrm{D} + 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & \rho\,\mathrm{D} + 1 \end{bmatrix}$$

As noted above, the second row of *Ext1* [1] corresponds to a zero row modulo the system equations given by R, so that we have to integrate two torsion elements. Taking the system equations into account once more, we check that these two torsion elements are in fact, up to the sign, equal modulo the rows of R:

```
> SyzygyModule(linalg[stackmatrix](Ext1[2], R), Alg);
```

$$\begin{bmatrix} 1 & 0 & 1 & -2 & 0 & -2 \\ 0 & 1 & 0 & 0 & -1 & -2 \\ 0 & 0 & \rho\,\mathrm{D} + 1 & -2\,\rho\,\mathrm{D} & -1 & -2\,\rho\,\mathrm{D} - 3 \end{bmatrix}$$

The rows of the preceding result generate all linear relations that hold for the union of the rows of *Ext1* [2] and R. So for each linear relation, the ith column gives the coefficient of the ith row of *Ext1* [2], if $1 \leq i \leq 3$ and the $(3 + i)$th column gives the coefficient of the ith row of R, $i = 1, 2, 3$. Hence, we see from the first row of the preceding result, that the two torsion elements given by the first and the third row of *Ext1* [2] are equal up to the sign modulo the rows of R.

The command that takes all this into account and performs the integration of the torsion elements is *IntTorsion*:

```
> T := IntTorsion(R, Alg);
```

$$T := \left[\begin{bmatrix} 1 & 1 \\ 0 & \rho\,\mathrm{D} + 1 \end{bmatrix}, \begin{bmatrix} -\frac{\_C1}{\rho} \\ \frac{\_C1}{\rho} \end{bmatrix}, \begin{bmatrix} 1 & \lambda + \mu & -1 & -2\,\mu \\ 0 & \lambda\,\rho\,\mathrm{D} + \mu\,\rho\,\mathrm{D} - 2\,\mu & 0 & 4\,\mu\,\rho\,\mathrm{D} + 6\,\mu \end{bmatrix} \right]$$

If we denote the two torsion elements defined by the first and third row of *Ext1* [2] by θ_1 and θ_2, we interpret the result of *IntTorsion* as follows: $T[1]$ gives the relations satisfied by θ_1 and θ_2 in M, namely, $\theta_1 = -\theta_2$ and $(\rho\,\mathrm{D} + 1)\,\theta_2 = 0$. These equations are solved to obtain the function in $T[2]$. Finally, $T[3]$ gives the definitions of θ_1 and θ_2 in terms of the system variable z.

The command *AutonomousElements* gives the same information as *IntTorsion*, but expresses the equations and torsion elements in terms of the system variables $\theta(\rho)$, $\sigma(\rho)$, $\mathrm{K}(\rho)$, $\mathrm{G}(\rho)$:

```
> AutonomousElements(R, [theta(rho),sigma(rho),K(rho),G(rho)], Alg);
```

$$\left[\left[\begin{array}{c}\theta_1(\rho)+\theta_2(\rho)=0\\ \theta_2(\rho)+\rho\,(\frac{d}{d\rho}\,\theta_2(\rho))=0\end{array}\right], \left[\begin{array}{c}\theta_1=-\dfrac{\_C1}{\rho}\\ \theta_2=\dfrac{\_C1}{\rho}\end{array}\right],\right.$$

$$\left.\left[\begin{array}{c}\theta_1=\theta(\rho)+\sigma(\rho)\,\lambda+\sigma(\rho)\,\mu-\mathrm{K}(\rho)-2\,\mu\,\mathrm{G}(\rho)\\ \theta_2=-2\,\sigma(\rho)\,\mu+(\frac{d}{d\rho}\,\sigma(\rho))\,\lambda\,\rho+(\frac{d}{d\rho}\,\sigma(\rho))\,\mu\,\rho+6\,\mu\,\mathrm{G}(\rho)+4\,\mu\,\rho\,(\frac{d}{d\rho}\,\mathrm{G}(\rho))\end{array}\right]\right]$$

Now, we need to find a particular solution of the inhomogeneous system $R\,z=\tau$, namely, *Ext1* $[2]\ z=\tau$. As $M/\ \mathrm{t}(M)$ is a torsion-free left A_1-module, it is also a stably free left A_1-module (A_1 is a hereditary ring). This result can be easily checked by noticing that *Ext1* $[2]$ has a full row rank and admits a right-inverse S:

```
> SyzygyModule(Ext1[2], Alg);
```

$$\mathrm{INJ}(3)$$

```
> S := RightInverse(Ext1[2], Alg);
```

$$S:=\left[\begin{array}{ccc}1 & \dfrac{(\rho\,\mathrm{D}+1)\,(\lambda+\mu)}{2\,\mu+\lambda} & -\dfrac{2\,\mu\,\rho\,\mathrm{D}+\mu+2\,\lambda\,\rho\,\mathrm{D}+\lambda}{3\,\lambda+7\,\mu}\\ 0 & -\dfrac{3+2\,\rho\,\mathrm{D}}{2\,\mu+\lambda} & \dfrac{4\,(\rho\,\mathrm{D}+1)}{3\,\lambda+7\,\mu}\\ 0 & -\dfrac{2\,(\lambda\,\rho\,\mathrm{D}+\lambda+\mu\,\rho\,\mathrm{D})}{2\,\mu+\lambda} & \dfrac{2\,(\lambda-\mu+2\,\lambda\,\rho\,\mathrm{D}+2\,\mu\,\rho\,\mathrm{D})}{3\,\lambda+7\,\mu}\\ 0 & \dfrac{\lambda\,\rho\,\mathrm{D}+\mu\,\rho\,\mathrm{D}-2\,\mu}{2\,\mu\,(2\,\mu+\lambda)} & -\dfrac{-\lambda-5\,\mu+2\,\lambda\,\rho\,\mathrm{D}+2\,\mu\,\rho\,\mathrm{D}}{2\,\mu\,(3\,\lambda+7\,\mu)}\end{array}\right]$$

We can easily check that S is a right-inverse of *Ext1* $[2]$, i.e., *Ext1* $[2]\ S=I_3$, by computing:

```
> Mult(Ext1[2],S,Alg);
```

$$\left[\begin{array}{ccc}1 & 0 & 0\\ 0 & 1 & 0\\ 0 & 0 & 1\end{array}\right]$$

Then, we obtain that a particular solution y of the inhomogeneous system *Ext1* $[2]\ y=\tau$ is defined by $y=S\,\tau$, where τ is the following vector:

```
> tau := vector([T[2][1,1], 0, T[2][2,1]]);
```

$$\tau:=\left[-\frac{\_C1}{\rho},\ 0,\ \frac{\_C1}{\rho}\right]$$

Therefore, we obtain that a particular solution y of *Ext1* $[2]\ y=\tau$ defined by:

```
> ApplyMatrix(S, tau, Alg);
```

$$\left[\begin{array}{c}-\dfrac{2\,\_C1\,(\lambda+3\,\mu)}{\rho\,(3\,\lambda+7\,\mu)}\\ 0\\ -\dfrac{2\,\_C1\,(\lambda+3\,\mu)}{\rho\,(3\,\lambda+7\,\mu)}\\ \dfrac{\_C1}{2\,\rho\,\mu}\end{array}\right]$$

In fact, the previous particular solution can be directly obtained by using *ParticularSolution*:

```
> ParticularSolution(R, Alg);
```

$$\left[\left[\begin{array}{cccc} 1 & \lambda+\mu & -1 & -2\,\mu \\ 0 & -\lambda-2\,\mu & \rho\,\mathrm{D}+1 & 4\,\mu\,\rho\,\mathrm{D}+4\,\mu \\ 0 & \lambda\,\rho\,\mathrm{D}+\mu\,\rho\,\mathrm{D}-2\,\mu & 0 & 4\,\mu\,\rho\,\mathrm{D}+6\,\mu \end{array}\right], \left[\begin{array}{c} -\dfrac{2\,\_C1\,(\lambda+3\,\mu)}{\rho\,(3\,\lambda+7\,\mu)} \\ 0 \\ -\dfrac{2\,\_C1\,(\lambda+3\,\mu)}{\rho\,(3\,\lambda+7\,\mu)} \\ \dfrac{\_C1}{2\,\rho\,\mu} \end{array}\right], \left[\begin{array}{c} -\dfrac{\_C1}{\rho} \\ 0 \\ \dfrac{\_C1}{\rho} \end{array}\right]\right]$$

We note that the second matrix is exactly the solution y obtained previously.

Finally, we can glue the parametrization of the torsion-free part $M\ /\ \mathrm{t}(M)$ of M with the integration of the torsion elements in order to obtain a parametrization of M. Indeed, we have

$$R\,z=0 \iff R''\,\tau=0 \quad \text{and} \quad Ext1[2]\,z=\tau \iff z=S\,\tau+Ext1[3]\,\xi_1.$$

This parametrization can be directly obtained by using *Parametrization*:

```
> P := Parametrization(R, Alg);
```

$$\begin{array}{l} P := \\ \Big[-2(\_C1\,\lambda+3\,\_C1\,\mu+3\,\mu^2\,\rho\,\xi_1(\rho)\,\lambda+7\,\mu^3\,\rho\,\xi_1(\rho)-3\,\mu\,\rho^3\,\%2\,\lambda^2 \\ -10\,\mu^2\,\rho^3\,\%2\,\lambda-7\,\mu^3\,\rho^3\,\%2-9\,\mu\,\rho^2\,\%1\,\lambda^2-30\,\mu^2\,\rho^2\,\%1\,\lambda-21\,\mu^3\,\rho^2\,\%1)/(\\ (3\,\lambda+7\,\mu)\,\rho)\Big] \\ \big[-2\,\mu\,(3\,\xi_1(\rho)+2\,\%2\,\rho^2+7\,\%1\,\rho)\big] \\ \Big[-2(\_C1\,\lambda+3\,\_C1\,\mu+9\,\mu\,\rho\,\xi_1(\rho)\,\lambda^2+27\,\mu^2\,\rho\,\xi_1(\rho)\,\lambda+14\,\mu^3\,\rho\,\xi_1(\rho) \\ +20\,\mu^2\,\rho^3\,\%2\,\lambda+14\,\mu^3\,\rho^3\,\%2+6\,\mu\,\rho^3\,\%2\,\lambda^2+54\,\mu^2\,\rho^2\,\%1\,\lambda+28\,\mu^3\,\rho^2\,\%1 \\ +18\,\mu\,\rho^2\,\%1\,\lambda^2)/((3\,\lambda+7\,\mu)\,\rho)\Big] \\ \left[\dfrac{1}{2}\,\dfrac{\_C1-4\,\mu^2\,\rho\,\xi_1(\rho)+2\,\%2\,\lambda\,\rho^3\,\mu+2\,\mu^2\,\rho^3\,\%2+4\,\%1\,\lambda\,\rho^2\,\mu}{\mu\,\rho}\right] \\ \%1 := \frac{d}{d\rho}\,\xi_1(\rho) \\ \%2 := \frac{d^2}{d\rho^2}\,\xi_1(\rho) \end{array}$$

We can check that P is a parametrization of the system:

```
> simplify(ApplyMatrix(R, P, Alg));
```

$$\left[\begin{array}{c} 0 \\ 0 \\ 0 \end{array}\right]$$

Therefore, we obtain the following parametrization of the system $R\,z=0$ which contains some autonomous elements:

```
> evalm([[theta(rho)], [sigma(rho)], [K(rho)], [G(rho)]]) = evalm(P);
```

$$\begin{bmatrix} \theta(\rho) \\ \sigma(\rho) \\ \mathrm{K}(\rho) \\ \mathrm{G}(\rho) \end{bmatrix} =$$

$$\Big[-2(\_C1\,\lambda + 3\,\_C1\,\mu + 3\,\mu^2\,\rho\,\xi_1(\rho)\,\lambda + 7\,\mu^3\,\rho\,\xi_1(\rho) - 3\,\mu\,\rho^3\,\%2\,\lambda^2$$
$$- 10\,\mu^2\,\rho^3\,\%2\,\lambda - 7\,\mu^3\,\rho^3\,\%2 - 9\,\mu\,\rho^2\,\%1\,\lambda^2 - 30\,\mu^2\,\rho^2\,\%1\,\lambda - 21\,\mu^3\,\rho^2\,\%1)/($$
$$(3\,\lambda + 7\,\mu)\,\rho)\Big]$$
$$\big[-2\,\mu\,(3\,\xi_1(\rho) + 2\,\%2\,\rho^2 + 7\,\%1\,\rho)\big]$$
$$\Big[-2(\_C1\,\lambda + 3\,\_C1\,\mu + 9\,\mu\,\rho\,\xi_1(\rho)\,\lambda^2 + 27\,\mu^2\,\rho\,\xi_1(\rho)\,\lambda + 14\,\mu^3\,\rho\,\xi_1(\rho)$$
$$+ 20\,\mu^2\,\rho^3\,\%2\,\lambda + 14\,\mu^3\,\rho^3\,\%2 + 6\,\mu\,\rho^3\,\%2\,\lambda^2 + 54\,\mu^2\,\rho^2\,\%1\,\lambda + 28\,\mu^3\,\rho^2\,\%1$$
$$+ 18\,\mu\,\rho^2\,\%1\,\lambda^2)/((3\,\lambda + 7\,\mu)\,\rho)\Big]$$
$$\left[\frac{1}{2}\,\frac{\_C1 - 4\,\mu^2\,\rho\,\xi_1(\rho) + 2\,\%2\,\lambda\,\rho^3\,\mu + 2\,\mu^2\,\rho^3\,\%2 + 4\,\%1\,\lambda\,\rho^2\,\mu}{\mu\,\rho}\right]$$
$$\%1 := \tfrac{d}{d\rho}\,\xi_1(\rho)$$
$$\%2 := \tfrac{d^2}{d\rho^2}\,\xi_1(\rho)$$

If we only consider smooth functions θ_1 and θ_2 at the origin, i.e., at $\rho = 0$, we then obtain $C_1 = 0$. This is the case considered by J. Hadamard in his paper. In that case, we obtain $\theta_1 = \theta_2 = 0$, and thus, the parametrization of the system becomes $z = \mathit{Ext1}[3]\ \xi_1$, i.e., it is the parametrization of the torsion-free left A_1-module $M/\ \mathrm{t}(M)$.

Now, instead of using the Weyl algebra, we represent the linear system of the rolling sphere by means of a matrix over the Ore algebra containing the Euler operator $E = \rho\ \mathrm{D}$.

We note that J. Hadamard exactly follows this approach.

```
> Alg2 := DefineOreAlgebra(euler=[E,rho], polynom=[rho], comm=[lambda,mu]):
```

The system matrix is now:

```
> R2 := evalm([[E+1/2, ((lambda+mu)/2)*(E-1), 1/2, 0],
> [2*E, -(3*lambda+2*mu), E+3, 0], [-E, lambda, -1, 2*mu*(E+1)]]);
```

$$\mathit{R2} := \begin{bmatrix} E + \frac{1}{2} & \frac{(\lambda+\mu)\,(E-1)}{2} & \frac{1}{2} & 0 \\ 2\,E & -3\,\lambda - 2\,\mu & E+3 & 0 \\ -E & \lambda & -1 & 2\,\mu\,(E+1) \end{bmatrix}$$

The corresponding linear system of equations is written as follows:

```
> ApplyMatrix(R2, [theta(rho),sigma(rho),K(rho),G(rho)], Alg2)=
> evalm([[0],[0],[0]]);
```

$$\begin{bmatrix} \frac{1}{2}\,\theta(\rho) + \rho\,(\frac{d}{d\rho}\,\theta(\rho)) - \frac{1}{2}\,\lambda\,\sigma(\rho) - \frac{1}{2}\,\sigma(\rho)\,\mu + \frac{1}{2}\,\rho\,(\frac{d}{d\rho}\,\sigma(\rho))\,\lambda + \frac{1}{2}\,\rho\,(\frac{d}{d\rho}\,\sigma(\rho))\,\mu + \frac{1}{2}\,\mathrm{K}(\rho) \\ 2\,\rho\,(\frac{d}{d\rho}\,\theta(\rho)) - 3\,\lambda\,\sigma(\rho) - 2\,\sigma(\rho)\,\mu + 3\,\mathrm{K}(\rho) + \rho\,(\frac{d}{d\rho}\,\mathrm{K}(\rho)) \\ -\rho\,(\frac{d}{d\rho}\,\theta(\rho)) + \lambda\,\sigma(\rho) - \mathrm{K}(\rho) + 2\,\mu\,\mathrm{G}(\rho) + 2\,\mu\,\rho\,(\frac{d}{d\rho}\,\mathrm{G}(\rho)) \end{bmatrix}$$
$$= \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

We apply an involution of the Ore algebra *Alg2* to the matrix *R2*:

```
> R_adj2 := Involution(R2, Alg2);
```

$$R\_adj2 := \begin{bmatrix} -E+\frac{1}{2} & -2\,E & E \\ -\frac{1}{2}\,\lambda\,E-\frac{1}{2}\,\lambda-\frac{1}{2}\,\mu\,E-\frac{1}{2}\,\mu & -3\,\lambda-2\,\mu & \lambda \\ \frac{1}{2} & -E+3 & -1 \\ 0 & 0 & -2\,\mu\,E+2\,\mu \end{bmatrix}$$

Let us check again whether or not the system is parametrizable.

```
> Ext2 := Exti(R_adj2, Alg2, 1);
```

$$Ext2 := \left[\begin{bmatrix} E+1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & E+1 \end{bmatrix}, \begin{bmatrix} 1 & \lambda+\mu & -1 & -2\,\mu \\ 0 & -\lambda-2\,\mu & E+1 & 4\,\mu+4\,\mu\,E \\ 0 & \lambda\,E+\mu\,E-2\,\mu & 0 & 6\,\mu+4\,\mu\,E \end{bmatrix}, \begin{bmatrix} 4\,\mu\,\lambda\,E+2\,\lambda\,\mu\,E^2-2\,\mu^2+2\,E^2\,\mu^2+4\,\mu^2\,E \\ -10\,\mu\,E-4\,\mu\,E^2-6\,\mu \\ -4\,\lambda\,\mu\,E^2-8\,\mu\,\lambda\,E-6\,\mu\,\lambda-4\,\mu^2-4\,E^2\,\mu^2-4\,\mu^2\,E \\ \lambda\,E^2+\lambda\,E-2\,\mu-\mu\,E+\mu\,E^2 \end{bmatrix} \right]$$

We find the same generators for the torsion submodule $\mathrm{t}(M)$ of M as previously. Using *TorsionElements*, we can find the following equivalent system for the torsion elements:

```
> TorsionElements(R2, [theta(rho),sigma(rho),K(rho),G(rho)], Alg2);
```

$$\left[\begin{bmatrix} \theta_1(\rho)+\rho\,(\frac{d}{d\rho}\,\theta_1(\rho))=0 \\ \theta_3(\rho)+\rho\,(\frac{d}{d\rho}\,\theta_3(\rho))=0 \end{bmatrix}, \right.$$

$$\left. \begin{bmatrix} \theta_1(\rho)=\theta(\rho)+\lambda\,\sigma(\rho)+\sigma(\rho)\,\mu-\mathrm{K}(\rho)-2\,\mu\,\mathrm{G}(\rho) \\ \theta_3(\rho)=-2\,\sigma(\rho)\,\mu+\rho\,(\frac{d}{d\rho}\,\sigma(\rho))\,\lambda+\rho\,(\frac{d}{d\rho}\,\sigma(\rho))\,\mu+6\,\mu\,\mathrm{G}(\rho)+4\,\mu\,\rho\,(\frac{d}{d\rho}\,\mathrm{G}(\rho)) \end{bmatrix} \right]$$

The previous torsion elements can be integrated by using *AutonomousElements*.

```
> AutonomousElements(R2, [theta(rho),sigma(rho),K(rho),G(rho)], Alg2);
```

$$\left[\begin{bmatrix} \theta_1(\rho)+\theta_2(\rho)=0 \\ \theta_2(\rho)+\rho\,(\frac{d}{d\rho}\,\theta_2(\rho))=0 \end{bmatrix}, \begin{bmatrix} \theta_1=-\frac{\_C1}{\rho} \\ \theta_2=\frac{\_C1}{\rho} \end{bmatrix}, \right.$$

$$\left. \begin{bmatrix} \theta_1=\theta(\rho)+\lambda\,\sigma(\rho)+\sigma(\rho)\,\mu-\mathrm{K}(\rho)-2\,\mu\,\mathrm{G}(\rho) \\ \theta_2=-2\,\sigma(\rho)\,\mu+\rho\,(\frac{d}{d\rho}\,\sigma(\rho))\,\lambda+\rho\,(\frac{d}{d\rho}\,\sigma(\rho))\,\mu+6\,\mu\,\mathrm{G}(\rho)+4\,\mu\,\rho\,(\frac{d}{d\rho}\,\mathrm{G}(\rho)) \end{bmatrix} \right]$$

It is known that the torsion elements are in one-to-one correspondence with the first integrals of the system. We can compute the corresponding first integral by using *FirstIntegral*:

```
> V := FirstIntegral(R2, [theta(rho),sigma(rho),K(rho),G(rho)], Alg2);
```

$$V := -\_C1\,\rho\,\theta(\rho)-\_C1\,\rho\,\sigma(\rho)\,\lambda-\_C1\,\rho\,\sigma(\rho)\,\mu+\_C1\,\rho\,\mathrm{K}(\rho)+2\,\mu\,\_C1\,\rho\,\mathrm{G}(\rho)$$

We note that the parametrization *Ext2*[3] of the torsion-free left A_1-module $M/\,\mathrm{t}(M)$ is exactly the one obtained by J. Hadamard in (38) of his paper up to the multiplication by the factor of $-1/(2\,\mu)$.

```
> Q := map(a->a/(-2*mu), map(a->map(b->collect(b, E), a), map(factor, Ext2[3])));
```

$$Q := \begin{bmatrix} -(\lambda+\mu)\,E^2-(2\,\lambda+2\,\mu)\,E+\mu \\ (2\,E+3)\,(E+1) \\ (2\,\lambda+2\,\mu)\,E^2+(2\,\mu+4\,\lambda)\,E+2\,\mu+3\,\lambda \\ -\dfrac{(E+1)\,((\lambda+\mu)\,E-2\,\mu)}{2\,\mu} \end{bmatrix}$$

Finally, we can directly compute the parametrization of the whole system $R\,z=0$ by using *Parametrization*.

```
> P2 := Parametrization(R2, Alg2);
```

$$P2 :=$$

$$\Big[-2(\_C1\,\lambda+3\,\_C1\,\mu+3\,\mu^2\,\rho\,\xi_1(\rho)\,\lambda+7\,\mu^3\,\rho\,\xi_1(\rho)-3\,\mu\,\rho^3\,\%2\,\lambda^2$$
$$-10\,\mu^2\,\rho^3\,\%2\,\lambda-7\,\mu^3\,\rho^3\,\%2-9\,\mu\,\rho^2\,\%1\,\lambda^2-30\,\mu^2\,\rho^2\,\%1\,\lambda-21\,\mu^3\,\rho^2\,\%1)/($$
$$(3\,\lambda+7\,\mu)\,\rho)\Big]$$

$$\big[-2\,\mu\,(3\,\xi_1(\rho)+2\,\%2\,\rho^2+7\,\%1\,\rho)\big]$$

$$\Big[-2(\_C1\,\lambda+3\,\_C1\,\mu+9\,\mu\,\rho\,\xi_1(\rho)\,\lambda^2+27\,\mu^2\,\rho\,\xi_1(\rho)\,\lambda+14\,\mu^3\,\rho\,\xi_1(\rho)$$
$$+20\,\mu^2\,\rho^3\,\%2\,\lambda+14\,\mu^3\,\rho^3\,\%2+6\,\mu\,\rho^3\,\%2\,\lambda^2+54\,\mu^2\,\rho^2\,\%1\,\lambda+28\,\mu^3\,\rho^2\,\%1$$
$$+18\,\mu\,\rho^2\,\%1\,\lambda^2)/((3\,\lambda+7\,\mu)\,\rho)\Big]$$

$$\left[\frac{1}{2}\,\frac{\_C1-4\,\mu^2\,\rho\,\xi_1(\rho)+2\,\%2\,\lambda\,\rho^3\,\mu+2\,\mu^2\,\rho^3\,\%2+4\,\%1\,\lambda\,\rho^2\,\mu}{\mu\,\rho}\right]$$

$$\%1 := \tfrac{d}{d\rho}\,\xi_1(\rho)$$

$$\%2 := \tfrac{d^2}{d\rho^2}\,\xi_1(\rho)$$

```
> simplify(evalm(P-P2));
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

Therefore, we obtain the same parametrization of the system independently of whether it is defined over the Weyl algebra A_1 or over the Ore algebra *Alg2* generated by the Euler operator $E=\rho\,\mathrm{D}$.